
*
*     C0 .2uF   ___||__
*             _|_  ||  |
*             \\\      |  -> C_equiv = -.19uF @ IABC=1000uA
*    VIN      R0       | :
*     ______/\  /\  /\_|_:_________________  OUT
*    |        \/  \/     :                 |
*    |                    ->  IAB = 1mA->0 |
*   _|_           __________|\             | |\
*  /_  \        _|_     INN | \/\/\________|_| \
* // \  \       \\\   ______| /\/\/          | /| VOUT
* \   \//             |   + |/        B_BUF1 |/ |
*  \___/           INP|    B_OTA1               |
*   _|_               |                         |
*   ///       R1      |               R2        |
*         _/\  /\  /\_|_________/\  /\  /\______|
*       _|_  \/  \/      |        \/  \/        |
*       ///              |____||________________|
*                             || C1
*
                               

VIN       VIN        0     SIN(   0    10m   1000   )  AC 10m
R0        VIN        OUT   10K
R1        INP        0     100
R2        VOUT       INP   1000k
C1        VOUT       INP   .1u
C0        OUT        0     .2u
B_OTA1    OUT        0     I =   -1*v(VIABC)*tanh(( v(INP) )/.052)
B_BUF1    VOUT       0     V =    v(OUT)
V_Iabc    VIABC      0     PWL    ( 0 1m 10m 0 ) 

.control
tran      1u         10m   0      1u  
set       pensize =  2
plot      v(vin)     v(out)      
op
ac        dec        10    1      1000K 
plot      db(out)    -db(vin)     title Gain2OUT@IABC1000uA
alter     V_Iabc     dc =         10u
ac        dec        10    1      1000K  
plot      db(out)    -db(vin)     title Gain2OUT@IABC10uA
.endc
.end

=====================END_OF_SPICE============================

This code works with winspice.

Just as a Negative Resistor can be made to cancel out an 
external resistor, the same is true for building a Negative
Capacitor. In this case most of C0's external capacitance
is going to be canceled out. Again special care is taken
such that the negative capacitance never exceeds the 
external capacitance.



*
*     C0 .2uF   ___||__
*             _|_  ||  |
*             \\\      |  -> C_equiv = -.19uF @ IABC=1000uA
*    VIN      R0       | :
*     ______/\  /\  /\_|_:_________________  OUT
*    |        \/  \/     :                 |
*    |                    ->  IAB = 1mA->0 |
*   _|_           __________|\             | |\
*  /_  \        _|_     INN | \/\/\________|_| \
* // \  \       \\\   ______| /\/\/          | /| VOUT
* \   \//             |   + |/        B_BUF1 |/ |
*  \___/           INP|    B_OTA1               |
*   _|_               |                         |
*   ///       R1      |               R2        |
*         _/\  /\  /\_|_________/\  /\  /\______|
*       _|_  \/  \/      |        \/  \/        |
*       ///              |____||________________|
*                             || C1
*
        V_Iabc    VIABC      0     PWL    ( 0 1m 10m 0 ) 

At 1mA the negative capacitor is trying to make 
C0 look like a open circuit. As IABC decreases
the capacitance approaches the .2uF of C0.

At 10uA, 10kOhm into 0.2uF is around 70Hz. 



At 1000uA, the 3dB bandwidth has increased to
1530Hz. So the 0.22uF is effectively reduced
by a factor of 20. 


